

---

# MP 4 – A Recursive Descent Parser for MiniJava

CS 421 – Spring 2011  
Revision 1.0

**Assigned** February 7, 2011  
**Due** February 14, 2010 11:59pm  
**Extension** 48 hours (20% penalty)  
**Total Points** 40 (no extra credit)

---

## 1 Change Log

1.0 Initial Release.

## 2 Overview

Previously, you created a lexer for MiniJava. Now it is time to write the parser that will build the Abstract Syntax Tree of the input program. In this MP, we will write a recursive descent (top-down) parser for a subset of MiniJava. The next MP will involve writing a bottom-up parser.

## 3 Collaboration

Collaboration is NOT allowed on this assignment.

## 4 What to submit

In this handout you will find several exercises, labeled as **Exercise**. Studying these exercises is for your own benefit. They are meant to help you see important concepts and to provoke thought. You will NOT be graded for your answers to these exercises. Therefore, you will NOT submit any answers for them (of course, you are welcome to discuss these exercises with the course staff or your friends).

What you will submit for this MP is the implementation of a recursive descent parser for MiniJava. You will submit your source in a file named `mp4.ml` using the `handin` program as usual.

## 5 Original Grammar

The grammar of MiniJava is given below. For this MP, we are only parsing class and method declarations, so the grammar is a subset of the regular MiniJava grammar.

```

Program ::= (ClassDecl)+
ClassDecl ::= "class" <IDENTIFIER> ("extends" <IDENTIFIER>)? "{" VarDecl* MethodDecl* "}"
VarDecl ::= Type <IDENTIFIER> ";"
           | "static" Type <IDENTIFIER> ";"
MethodDecl ::= "public" Type <IDENTIFIER>
              "(" (Type <IDENTIFIER> ("," Type <IDENTIFIER>)* )? ")" "{" "}"
Type ::= Type "[" "]"
       | "boolean"
       | "float"
       | "int"
       | <IDENTIFIER>

```

## 6 Tokens

You will be given a correctly implemented lexer for your use. The lexer will tokenize the input, and output the list of lexemes. The token type is defined as below. Note that we have omitted several unneeded tokens.

```

type token =
  BOOLEAN | CLASS | EXTENDS | FLOAT | INT
  | PUBLIC | STATIC | LPAREN | RPAREN | LBRACE | RBRACE
  | LBRACK | RBRACK | SEMICOLON | COMMA | IDENTIFIER of string
  | ... (* Several other tokens not needed for this MP *)

```

Also note that we no longer have the EOF token; end-of-file is indicated by an empty list.

## 7 AST Structure

As the result of parsing, you will return an AST of the input program based on the following definition. Note that this is similar to the abstract syntax you saw in MP2, but with method bodies removed and a `var_kind` field added to `var_decl` to indicate Static vs. NonStatic variable declarations.

```

type program = Program of (class_decl list)

and class_decl = Class of id * id
                * (var_decl list)
                * (method_decl list)

and var_decl = Var of var_kind * exp_type * id

and var_kind = Static | NonStatic

and method_decl = Method of exp_type
                    * id
                    * ((exp_type * id) list)

and exp_type = ArrayType of exp_type
              | BoolType
              | IntType
              | FloatType
              | ObjectType of id

and id = string

```

### Exercise

Compare this AST structure to the grammar given in Section 5, and observe how the non-terminals and rules in the grammar map to the AST structures.

## 8 Modified Grammar

The grammar we gave you in Section 5 contains regular expression abbreviations such as + and ?, and also the Kleene star \*. We now factor out these for you to make the grammar more suitable for top-down parsing. The new grammar is given in Figure 1. You will use this grammar to write the parser. The original grammar is only for reference.

```
Program ::= ClassDecl Classes
Classes ::= ClassDecl Classes
         |  $\epsilon$ 
ClassDecl ::= "class" <IDENTIFIER> ClassA
ClassA ::= "extends" <IDENTIFIER> "{" ClassB
        | "{" ClassB
ClassB ::= "}"
        | "public" MethodDecl ClassC
        | VarDecl ClassB
ClassC ::= "}"
        | "public" MethodDecl ClassC
VarDecl ::= "static" Type <IDENTIFIER> ";"
         | Type <IDENTIFIER> ";"
MethodDecl ::= Type <IDENTIFIER> "(" MethodA
MethodA ::= ")" "{" "}"
         | Type <IDENTIFIER> MethodB
MethodB ::= ")" "{" "}"
         | "," Type <IDENTIFIER> MethodB
Type ::= SimpleType ArrayPart
SimpleType ::= "boolean"
            | "float"
            | "int"
            | <IDENTIFIER>
ArrayPart ::= "[" "]" ArrayPart
           |  $\epsilon$ 
```

Figure 1: The modified grammar.

### Exercise

Compare and contrast the new grammar (Figure 1) to the original one in Section 5. Make sure that you understand they define the same language.

### Exercise

Remember how the original grammar non-terminals and rules would map the AST datatype. Now try to match the rules of the new grammar to the AST structure. Is there still a direct mapping? Why/Why not?

## 9 Recursive Descent Parsing

To write a recursive descent (i.e. top-down) parser, the initial requirement is that the grammar should be top-down parsable. There are two important issues in a grammar one should pay attention to:

- Left-recursion. Rules in the form

$$A ::= Aa \mid \dots$$

- The FIRST test: It must be possible to determine what production to use by looking at the next input token.

These issues make a grammar inappropriate for recursive descent parsing (recall why). Such grammars have to be factored to become top-down parsable.

### Exercise

Examine the grammar given in Figure 1. Verify that it does not have left-recursion and it passes the FIRST test. Convince yourself that it is top-down parsable.

## 10 Writing the Parser

Here comes the fun part. You are now ready to write the recursive descent parser for MiniJava. Keep in mind that you will use the factored grammar given in Figure 1. Here are the instructions:

- Download `mp4grader.tar.gz`. This tarball contains all the files you need, including
  - The lexer
  - The definition of tokens and the abstract syntax (in `mp4common.mli`).
  - A skeleton file for you to start from (`mp4-skeleton.ml`).
  - Several test cases (`tests`)
- As always, extract the tarball, rename `mp4-skeleton.ml` to `mp4.ml` and start modifying the file. You will modify only the `mp4.ml` file, and submit this file only.
- Compile your solution with `make`. Run the `./grader` to see how well you do.
- Make sure to add several more test cases to the `tests` file.
- The following will allow you to run the solution parser interactively:

```
# #load "solution.cmo";;
# #load "minijavalex.cmo";;
# let parse s = Solution.parse (Minijavalex.get_all_tokens s);;
val parse : string -> Mp4common.program option = <fun>
# parse "class Dummy { }";;
- : Mp4common.program option =
Some (Mp4common.Program [Mp4common.Class ("Dummy", "", [], [])])
```

## Filling in the skeleton file

In the skeleton file you will notice that there is a function for each non-terminal of the grammar. A function receives a list of tokens. It then parses the part that it is supposed to recognize, and returns the abstract syntax tree of this part together with the rest of the tokens. For instance, `parse_ClassDecl`, given the tokens of the input

```
class A { } int i;
```

should return the following result

```
- : Mp4common.class_decl * Mp4common.token list =  
(Mp4common.Class ("A", "", [], []),  
 [Mp4common.INT; Mp4common.IDENTIFIER "i"; Mp4common.SEMICOLON])
```

Note that the first item of the returned pair is the AST for “`class A { }`”, and the second item is the remaining tokens for “`int i;`”. When given the tokens for “`notAClass`” as the input, a `Parse_failure` exception is returned.

The first function of the parser, `parse_program`, is implemented for you and copied below:

```
let rec parse_Program toks =  
  let (c, toks') = parse_ClassDecl toks  
  in let (cs, toks'') = parse_Classes toks'  
  in  
    (Program(c::cs), toks'')
```

Note how this implementation follows the `Program` rule of the grammar.

Finally, there is a top-level function called `parse` that receives a token list and returns the AST. This is the function we will interact with. It has also been implemented for you.

## Parse\_failure Exception

Note the following line at the top of the skeleton:

```
exception Parse_failure
```

This line defines an **exception** called `Parse_failure`, which is to be raised whenever your parser cannot proceed due to a bad token list. In OCaml, exceptions can be raised at any point, and cause control to be handed up to the nearest enclosing `try...with...` block.

For this assignment, the top-level function `parse` (which we have already defined for you) contains a `try...with...` block to handle the exception `Parse_failure` for you (so you never need to write a `try...with...` block yourself.)

## What should the functions return?

A `parse_X` function, when it can parse the input, is supposed to return the AST of the parsed tokens and the rest of the tokens (i.e. the tokens that it did not consume). For instance, the `parse_Program` function returns the `Program` that contains the list of classes. It also returns the tokens left after the sequence of classes. So a pair of values will be returned.

If the parsing of some rule cannot continue because the token list does not match any of the patterns it is expected to, then `raise Parse_failure`. The exception will be propagated all the way back up to the top-level `parse` function, which will return `None`.

Note that not every function corresponds to an AST structure as `Program` does. For example, `parse_Classes`, called from `parse_Program`, returns a list of AST's. For other functions, by looking at the grammar rules, you should figure out what to return; you may need to return a single AST, a list of AST's, or a tuple containing several things.

**Hint:** Constructing the correct AST for array types is a bit trickier than other parts. You may want to pass an extra argument to the `parse_ArrayPart` function (as we have in the skeleton), or implement a helper function to post-process its output.

## 11 The Assignment

Write a recursive descent parser for MiniJava according to the grammar given in Figure 1. You need to implement the parser functions for non-terminals as given in the skeleton file. The point distribution is given below (subject to change).

Be able to parse	Approx. points
classes with empty body	5
classes with fields only	10
classes with methods only	10
classes with fields and methods	10
Reject bad inputs	5
Total	40